
Liveness-to-safety

Objective
Encode the search for a fair path as a reachability problem.

How
Every solution [path] for the reachability problem should represent
a fair lasso-shape in the original system.

Single fairness encoding

Fair transition system M ..= 〈S, I, T, F 〉

• for x ∈ S, introduce a new symbol l x.

• The assignment of every such symbol l x is
non-deterministally chosen in the initial state and then it
never changes (FROZENVAR).

• The loopback state is lback ..=
∧

x∈S l x = x.

• Boolean symbol in l, initially false and in l′ = in l ∨ lback′.
• Boolean symbol fair l, initially false and
fair l′ = fair l ∨ (in l ∧ F )

Find path starting from I that reaches lback ∧ fair l.

Single fairness encoding: example

Modulo 8 counter

MODULE main
VAR
b0 : boolean;
b1 : boolean;
b2 : boolean;

ASSIGN
init(b0) := FALSE;
init(b1) := FALSE;
init(b2) := FALSE;
next(b0) := !b0;
next(b1) := (!b0 & b1) | (b0 & !b1);
next(b2) := ((b0 & b1) & !b2) | (!(b0 & b1) & b2);

DEFINE out := toint(b0) + 2 * toint(b1) + 4 * toint(b2);

LTLSPEC G F out != 2;

Single fairness encoding: solution

Add the following:

FROZENVAR
l_b0 : boolean;
l_b1 : boolean;
l_b2 : boolean;

DEFINE out_lback := toint(l_b0) + 2 * toint(l_b1) + 4 * toint(l_b2);

DEFINE lback := l_b0 = b0 & l_b1 = b1 & l_b2 = b2;
DEFINE fair := out = 2;

VAR
in_loop : boolean;
fair_loop : boolean;

INIT !in_loop & !fair_loop;
TRANS next(in_loop) = in_loop | next(lback);
TRANS next(fair_loop) = fair_loop | (in_loop & fair);

INVARSPEC !(fair_loop & lback);

Multiple fairness conditions

Q: How can we deal with multiple fairness conditions?

A1 Usual reduction to single fairness.
Requires to visit fairness conditions in a predefined order, can
cause very long loops even when a shorter one exists.

A2 Add one boolean symbol fair li for each fairness. Look for a
path that reaches l back ∧

∧
i fair li

Multiple fairness conditions encoding: example

Modulo 8 counters

MODULE main
VAR

c0 : counter();
c1 : counter();

LTLSPEC (F G c0.out != 2) | (F G c1.out != 4);

MODULE counter
VAR

b0 : boolean;
b1 : boolean;
b2 : boolean;

ASSIGN
init(b0) := FALSE;
init(b1) := FALSE;
init(b2) := FALSE;
next(b0) := !b0;
next(b1) := (!b0 & b1) | (b0 & !b1);
next(b2) := ((b0 & b1) & !b2) | (!(b0 & b1) & b2);

DEFINE out := toint(b0) + 2 * toint(b1) + 4 * toint(b2);

Multiple fairness conditions encoding: solution

See files in examples for both versions.

10 / 13



Make a transition system total

Q: Given a transition system M ..= 〈S, I, T 〉, can we define
Mt

..= 〈St, It, Tt〉, such that every path in M has a corresponding
path in Mt and Tt is total?

• St ..= S ∩ {err},

• It ..= I ∧ ¬err,

• Tt ..= ((¬err ∧ T ) → ¬err′) ∧ (err ∨ ¬T ) → err′.

If Mt |= (G¬err) ∧ φ then M |= φ.

Q: What’s the relation between the err states of Mt and the
deadlocks of M?
We are adding a lot of transitions!!

